cross database ownership - The server principal is not able to access the database under the current security context
The server principal "My_Company\sqldbslaveowner" is not able to
  access the database "JUNOCORE" under the current security context.


You should not need to EXECUTE AS OWNER in this scenario, and it will break as you can't cross databases when impersonating a database principal.  Impersonation and Ownership Chains are two separate mechanisms for delegation.

And the normal thing that people forget is that the end user must have database access to the target database.  Cross Database Ownership chains don't give a user access to a database, they only suppress permissions checks on objects in that database.

Here's a simple repro:

/*
revert
use master
drop database a
drop database b
drop login appuser 
*/
go
create database a
create database b

ALTER DATABASE a SET DB_CHAINING ON; 
ALTER DATABASE b SET DB_CHAINING ON; 
go

use b
go
create procedure foo
as
begin
  select 1 a;
end

go

use a
go
create procedure foo
as
begin
  exec b.dbo.foo;
end

go

create login appuser with password='P@Jhs9d!$12'
create user appuser for login appuser
grant exec on foo to appuser

go
execute as login='appuser'
  exec foo;
--fails
--Msg 916, Level 14, State 1, Procedure foo, Line 4 [Batch Start Line 37]
--The server principal "appuser" is not able to access the database "b" under the current security context.

go
revert

go
use b
create user appuser for login appuser
deny select,insert,update,execute to appuser --not strictly necessary
go
use a
execute as login='appuser'
  exec foo;
revert
--works

go

--`execute as owner` is incompatible with cross database ownership chains
create or alter procedure foo
with execute as owner
as
begin
  exec b.dbo.foo;
end
go

use a
execute as login='appuser'
  exec foo;
revert
--fails
--Msg 916, Level 14, State 1, Procedure foo, Line 5 [Batch Start Line 69]
--The server principal "NORTHAMERICA\dbrowne" is not able to access the database "b" under the current security context.